Deux suites finies des nombres à la fois triangulaires et pyramidaux

triangulaires et pyramidaux à base carrée.

La suite A039596 de l'encyclopédie en ligne OEIS donne la liste des nombres qui sont à la fois triangulaires et pyramidaux : 1, 55, 91, 208335.
Nous nous proposons de les retrouver.
On peut immédiatement penser que le nombre triangulaire général $T(n)$ correspond à une surface et est un polynôme en n de degré 2 tandis que le nombre pytamidal $P(m)$ qui correspond à un volume, est un polynôme de degré 3 en m. Ce qui fait qu'en égalisant on obtiendra une équation d'une courbe elliptique.



Nombres triangulaires *Nombres pyramidaux*

La suite (T) des nombres triangulaires est $T = 0, 1, 3, 6, 10, \ldots$ de terme général $T(n) = \sum_{k=0}^{n} k = \frac{n(n+1)}{2}$ et la suite (P) des nombres pyramidaux est $P = 0, 1, 5, 14, 30, \ldots$ de terme général $P(n) = \sum_{k=0}^{n} k^2 = \frac{n}{6}(n+1)(2n+1)$.
L'équation à résoudre en n et en m est $T(n) = P(m)$.

$$\begin{array}{rcl}
T(n) & = & P(m) \\
\frac{n(n+1)}{2} & = & \frac{m}{6}(m+1)(2m+1) \\
3(2n)(2n+2) & = & 2m(2m+2)(2m+1) \\
3(Y-1)(Y+1) & = & (X-1)X(X+1) \\
3Y^2 - 3 & = & X^3 - X \\
81Y^2 & = & 27X^3 - 27X + 81 \\
y^2 & = & x^3 - 9x + 81
\end{array}$$

Une première mise en forme a pour but de préparer l'utilisation de l'identité remarquable $(a-b)(a+b) = a^2 - b^2$.

Le changement de variables $Y = 2n + 1$, $X = 2m + 1$ aide à simplifier les calculs et nous conduit à effectuer un second changement de variables $y = 9Y$ et $x = 3X$ qui nous donne finalement une équation de Weierstrass de la forme $y^2 = x^3 + ax + b$.
Si m et n sont entiers, alors $x = 3(2m+1)$ et $y = 9(2n+1)$ le sont aussi, comme $m = \frac{x-3}{6}$ et $n = \frac{y-9}{18}$, la réciproque n'rest pas nécessairement toujours vraie, les valeurs de m et n obtenues sont rationnelles mais a priori rien ne garantit qu'elles soient entières ou même positives.
L'ensemble des solutions de $T(n) = P(m)$ (1) est donc l'ensemble des couples de deux nombres entiers positifs n et m obtenus à partir des solutions entières de $y^2 = x^3 - 9x + 81$ (2).

Les solutions de (2) se trouvent dans les tables de Cremona, on peut aussi les déterminer à l'aide d'un logiciel de théorie des nombres, celui-ci est `mwrank`, intégré dans `sage`. En voici l'exécution :

```
jp@abc:~$ sage
----------------------------------------------------------------------
| Sage Version 4.0.1, Release Date: 2009-06-06                       |
| Type notebook() for the GUI, and license() for information.        |
----------------------------------------------------------------------
E=EllipticCurve([0,0,0,-9,81]);
F=E.minimal_model();
p=F.S_integral_points([2,3]);
A=F.isomorphism_to(E);
q=[A(P) for P in p];
s=[((x-3)/6 ,(y-9)/18) for (x,y,z) in q];
s=[(x,y) for (x,y) in s if x.is_integral() and y.is_integral()];
s
sage: E=EllipticCurve([0,0,0,-9,81]);
sage: F=E.minimal_model();
sage: p=F.S_integral_points([2,3]);
sage: A=F.isomorphism_to(E);
sage: q=[A(P) for P in p];
sage: s=[((x-3)/6 ,(y-9)/18) for (x,y,z) in q];
sage: s=[(x,y) for (x,y) in s if x.is_integral() and y.is_integral()];
sage: s
[(-1, 0), (0, 0), (1, 1), (5, 10), (6, 13), (85, 645)]
sage:
```

Il ne reste plus qu'à calculer $T(0) = 0$, $T(1) = 1$, $T(10) = 55$, $T(13) = 91$, $T(85) = 208335$ pour obtenir le résultat.
NB. En France l'usage veut que 0 soit le plus petit entier positif (et non 1), on préfèrera sans doute faire débuter la suite par $T(0)$ et non par $T(1)$.

Nombres à la fois triangulaires et tétrahédraux.

Les nombres triangulaires sont décrits au paragraphe précédent. Les nombres tétrahédraux ou pyramidaux à base triangulaire sont les sommes des nombres triangulaires.
La suite (T) des nombres triangulaires est $T = 0, 1, 3, 6, 10, 15, 21, 28, 36, 45\ldots$ de terme général $T(n) = \sum_{k=0}^{n} k = \frac{n(n+1)}{2}$ et la suite (H) des nombres tétrahédraux est $H = 0, 1, 4, 10, 25, 46, 74, 110, 155\ldots$ de terme général $H(n) = \sum_{k=0}^{n} \frac{k(k+1)}{2} = \frac{n(n+1)(n+2)}{6}$.



Nombres triangulaires et nombres tétrahédraux

L'équation à résoudre en n et en m est cette fois $T(n) = H(m)$.

$$\begin{aligned}
T(n) &= H(m) \\
\frac{n(n+1)}{2} &= \frac{m(m+1)(m+2)}{6} \\
3(2n)(2n+2) &= 4(m)(m+1)(m+2) \\
3(Y-1)(Y+1) &= 4(X-1)X(X+1) \\
3(Y^2-1) &= 4(X^3-X) \\
81 \times 16(Y^2-1) &= 27 \times 64(X^3-X) \\
36^2Y^2 - 36^2 &= 12^3X^3 - 12^3X \\
(36Y)^2 &= (12X)^3 - 12^2(12X) + 36^2 \\
y^2 &= x^3 - 144x + 1296
\end{aligned}$$

Les changements de variables sont d'abord $Y = 2n+1$, $X = m+1$ puis $y = 36Y$, $x = 12X$.
Finalement $y = 36(2n + 1)$, $x = 12(m + 1)$ et inversement $n = \frac{y}{72} - \frac{1}{2}$, $m = \frac{x}{12} - 1$.

Le script correspondant à ce problème des nombres triangulaires et tétrahédraux, à exécuter dans Sage est

```
E=EllipticCurve([0,0,0,-144,16*81]);
F=E.minimal_model();
p=F.S_integral_points([2,3]);
A=F.isomorphism_to(E);
q=[A(P) for P in p];
s=[((x-12)/12 ,(y-36)/72) for (x,y,z) in q];
s=[(x,y) for (x,y) in s if x.is_integral() and y.is_integral()];
s
```

Les solutions obtenues à ce moment sont

```
[(-2, 0), (-1, 0), (0, 0), (1, 1), (3, 4), (8, 15), (20, 55), (34, 119)]
```

Les nombres sont $H(0) = T(0) = 0$, $H(1) = T(1) = 1$, $H(3) = T(4) = 10$, $H(8) = T(15) = 120$, $H(20) = T(55) = 1540$ et $H(34) = T(119) = 7140$

Déc. 2009, et mars 2010 Jean-Paul Davalan — `jeux et mathématiques` —

`http://linenn.davalan.eu/pdf/pyram.pdf`

`http://jeux-et-mathematiques.davalan.org/mots/suites/weierstrass/pyram.pdf`